Variational Principle in Practice

Recipe:

1. Define our system (Hamiltonian)
2. Propose trial wave function family (parametrized wave function, basis set)
3. Minimize parameters

Harmonic Oscilator

Ground State

Trial wave function family

```
γ[x_, α_] := Exp[-x^2 α]
Manipulate[Plot[γ[x, α], {x, -10, 10}], {α, -3, 3}]
```



```
Integrate[ɣ[x, α] ɣ[x, α], {x, -Infinity, Infinity}, Assumptions → α > 0]
```

$$\frac{\sqrt{\frac{\pi}{2}}}{\sqrt{\alpha}}$$

```
ψ[x_, α_] := Exp[-x^2 α] / Sqrt[Sqrt[π/2] / Sqrt[α]]
```

Hψ

```
H[x_, ψ_] := -ℏ^2 / (2 m) D[ψ, {x, 2}] + 1 / 2 m x^2 ω^2 ψ
```

```
H[x, ψ[x, α]]
```

$$\frac{e^{-x^2 \alpha}\, m\, x^2\, \alpha^{1/4}\, \omega^2}{2^{3/4}\, \pi^{1/4}} - \frac{\alpha^{1/4} \left(-2\, e^{-x^2 \alpha}\, \alpha + 4\, e^{-x^2 \alpha}\, x^2\, \alpha^2\right) \hbar^2}{2^{3/4}\, m\, \pi^{1/4}}$$

<H>= <ψ|H|ψ>

```
Integrate[ψ[x, α] H[x, ψ[x, α]], {x, -Infinity, Infinity}, Assumptions → α > 0]
```

$$\frac{m\, \omega^2}{8\, \alpha} + \frac{\alpha\, \hbar^2}{2\, m}$$

```
H[α_] = m ω^2 / (8 α) + α ℏ^2 / (2 m)
```

$$\frac{m\, \omega^2}{8\, \alpha} + \frac{\alpha\, \hbar^2}{2\, m}$$

```
Solve[D[H[α], α] == 0, α]
```

$$\left\{\left\{\alpha \to -\frac{m\, \omega}{2\, \hbar}\right\}, \left\{\alpha \to \frac{m\, \omega}{2\, \hbar}\right\}\right\}$$

```
H[m ω / (2 ℏ)]
```

$$\frac{\omega\, \hbar}{2}$$

```
ψ[x, m ω / (2 ℏ)] // FullSimplify
```

$$\frac{e^{-\frac{m\, x^2\, \omega}{2\, \hbar}} \left(\frac{m\, \omega}{\hbar}\right)^{1/4}}{\pi^{1/4}}$$

(First) Excited State

```
ϰ[x_, α_] := x Exp[-x^2 α]
```

```
Manipulate[Plot[ϰ[x, α], {x, -5, 5}], {α, .1, 5}]
```




```
Integrate[ϰ[x, α] ϰ[x, α], {x, -Infinity, Infinity}, Assumptions → α > 0]
```

$$\frac{\sqrt{\frac{\pi}{2}}}{4\,\alpha^{3/2}}$$

```
ξ[x_, α_] := x Exp[-x^2 α] / Sqrt[Sqrt[π/2] / (4 α^(3/2))]
```

$$\xi[x\_,\alpha\_] := x\,\mathrm{Exp}[-x^2\alpha] \Big/ \mathrm{Sqrt}\left[\frac{\sqrt{\frac{\pi}{2}}}{4\,\alpha^{3/2}}\right]$$

```
Integrate[ξ[x, α] H[x, ξ[x, α]], {x, -Infinity, Infinity}, Assumptions → α > 0]
```

$$\frac{3\,m\,\omega^2}{8\,\alpha} + \frac{3\,\alpha\,\hbar^2}{2\,m}$$

$$\mathrm{Solve}\left[D\left[\frac{3\,m\,\omega^2}{8\,\alpha} + \frac{3\,\alpha\,\hbar^2}{2\,m},\ \alpha\right] == 0,\ \alpha\right]$$

$$\left\{\left\{\alpha \to -\frac{m\,\omega}{2\,\hbar}\right\},\ \left\{\alpha \to \frac{m\,\omega}{2\,\hbar}\right\}\right\}$$

$$\mathrm{HH}[\alpha\_] := \frac{3\,m\,\omega^2}{8\,\alpha} + \frac{3\,\alpha\,\hbar^2}{2\,m}$$

$$HH\left[\frac{m\,\omega}{2\,\hbar}\right]$$

$$\frac{3\,\omega\,\hbar}{2}$$

Reference: Quantum mechanics, Cohen (vol II, p. 1148)

Homework:

1. Read Cohen (vol II, p. 1148) together with the notebook
2. Try non-exact wave function that is proposed there.
3. Try another odd function for the first exited state (sin(x)Exp[-x^2α])
4. Computationa physics, Thijssen, chapter 3.
5. Introduction to Quantum mechanics, Griffiths, chapter 7.